QUASILOCAL CONSERVED CHARGES IN THE QUANTUM HIROTA MODEL (Lenart Zadnik, 2017)

Weyl algebra representation : W[{i, j}, {k, l}] is a tensor product of (u^i v^j) and (u^k v^l), where local basis elements u and
v satisfy uv = q vu, q being the root of unity quantization parameter of the Hirota model. In this code q = Exp[I 2 Pi/3], change accordingly. Note, that u^2 = u^-1 and similarly for v.

```
$RecursionLimit = 65 536;
$IterationLimit = 65 536;

(*TENSOR PRODUCT*)
TensW[A_, B_ + C_] := TensW[A, B] + TensW[A, C];
TensW[B_ + C_, A_] := TensW[B, A] + TensW[C, A];
TensW[A_, B_ * C_] := B * TensW[A, C];
TensW[B_ * C_, A_] := B * TensW[C, A];
TensW[W[A___], W[B___]] := W[A, B];
TensW[_, _] := 0;

(*WEYL MULTIPLICATION*)
MultW[W[], W[]] := W[];
MultW[W[A_List, C___], W[B_List, D___]] :=
  TensW[Exp[-I 2 Pi A[[2]] B[[1]] / 3] W[Mod[A + B, 3]], MultW[W[C], W[D]]];

(*ORDINARY MULTIPLICATION*)
Mult[S1_W, S2_W] := Expand[MultW[S1, S2]];
Mult[A_ + B_, C_] := Mult[A, C] + Mult[B, C];
Mult[A_, B_ + C_] := Mult[A, B] + Mult[A, C];
Mult[A_ * B_, C_] := A * Mult[B, C];
Mult[A_, B_ * C_] := B * Mult[A, C];
Mult[_, _] := 0;
Prod = Expand@Fold[Mult, #] &;
Com[A_, B_] := Mult[A, B] - Mult[B, A];

(*TRACE ON THE PHYSICAL SPACE*)
TraceP[W[X__]] := If[
   MemberQ[Flatten[{X}], _? (# > 0 &)],
   0, 3 ^ (Length[{X}])
  ];
TraceP[Plus[A_, B__]] := TraceP[A] + TraceP[Plus[B]];
TraceP[Plus[B__, A_]] := TraceP[Plus[B]] + TraceP[A];
TraceP[A_ * B___W] := A * TraceP[B] /; FreeQ[A, W];
TraceP[A___] := TraceP[Expand[A]];
TraceP[A___] := A /; FreeQ[A, W];

(*EXTENSION ON THE FULL HILBERT SPACE: n = NUMBER OF SITES,
site_List = SITES WITH NONTRIVIAL ACTION*)
Exten[n_, site_List] := (# /. W[X___] /; Length[{X}] == Length[site] :>
      W @@ Permute[Join[{X}, Table[{0, 0}, n - Length[site]]],
        Join[site, DeleteCases[Range[n], _? (MemberQ[site, #] &)]]]) &;
```

```
(*R-MATRIX FOR THE THIRD ROOT OF UNITY*)
R[K_] := W[{0, 0}, {0, 0}] +
   (K^2 - 1) / (K^2 Exp[I 2 Pi / 3] - Exp[-I 2 Pi / 3]) (W[{1, 1}, {1, 2}] + W[{2, 2}, {2, 1}]);

(*TIME PROPAGATION, n = CHAIN LENGTH (EVEN INTEGER)*)
U[n_, K_] := Prod[
   Join[
    Table[
     Exten[n, {2 k - 1, 2 k}][R[K]],
     {k, 1, n / 2}],
    Table[
     Exten[n, {2 k, Mod[2 k + 1, n, 1]}][R[K]],
     {k, 1, n / 2}]
   ]
  ];

(*LAX COMPONENTS*)
Table[
  L[i, j][x_] = {{0, 0}, {0, 0}},
  {i, 0, 2}, {j, 0, 2}
 ];
L[1, 0][x_] = {{1, 0}, {0, 0}};
L[2, 0][x_] = {{0, 0}, {0, 1}};
L[0, 1][x_] = {{0, x}, {0, 0}};
L[0, 2][x_] = {{0, 0}, {-x, 0}};

(*STAGGERED LAX COMPONENTS*)
Table[stagL[i, j, k, l][x_, K_] = L[k, l][x / K].L[i, j][x K],
  {i, 0, 2}, {j, 0, 2}, {k, 0, 2}, {l, 0, 2}];

(*CORRESPONDING SINGLET EIGENVALUE, x = SPECTRAL PARAMETER, K = SCALING PARAMETER*)
sing[x_, K_] := 1 + x^2 / K^2 + K^2 x^2 + x^4;

(*DOUBLE LAX COMPONENTS*)
Table[
  LL[ii, jj, kk, ll][x_, y_, K_] =
   FullSimplify[
     Sum[
      Exp[I 2 Pi ((i - ii) j + (k - kk) l) / 3]
       KroneckerProduct[stagL[i, j, k, l][x Exp[I Pi / 3], K], stagL[Mod[ii - i, 3],
          Mod[jj - j, 3], Mod[kk - k, 3], Mod[ll - l, 3]][y Exp[-I Pi / 3], K]],
      {i, 0, 2}, {j, 0, 2}, {k, 0, 2}, {l, 0, 2}
     ]
    ] / sing[x, K],
```

```
  {ii, 0, 2}, {jj, 0, 2}, {kk, 0, 2}, {ll, 0, 2}
 ];

(*QUASILOCAL CONSERVED CHARGES: n = NUMBER OF LATTICE SITES,
K = SCALING PARAMETER, x = SPECTRAL PARAMETER*)
X[x_, K_, n_] := Block[
   {basis = Tuples[{0, 1, 2}, 2 n], y},
   Table[
    dummyLL[i, j, k, l][x, y, K] = LL[i, j, k, l][x, y, K],
    {i, 0, 2}, {j, 0, 2}, {k, 0, 2}, {l, 0, 2}
   ];
   D[
     Sum[
      Tr[
        Dot @@ ((dummyLL @@ #)[x, y, K] & /@ Partition[basis[[j]], 4])
       ] ×
       W @@ Partition[Flatten[Reverse[Partition[basis[[j]], 4]]], 2],
      {j, 1, Length[basis]}
     ], y
    ] /. y → x
  ];

(*TEST*)
Chop[Com[X[4.7, N[1/Sqrt[3]], 6], U[6, N[1/Sqrt[3]]]]]
```